
	Proposed Solution
	A general description of the proposed solution is given here. Changes required in the text of the standard are given below.

Summary of the solution
Allow to add a default value to formal parameters. A default value may be added to any of the formal parameters (i.e. there is no restriction in the order of parameters with and without default values). When calling a parameterized object, actual values shall be provided for parameters without default values (called mandatory parameters from now on) and may be provided for parameters with default values (called non-mandatory parameters from now on). If an actual value is provided for a non-mandatory parameter, this value is used by the given instance of the object, otherwise the default value is used. Non-mandatory parameters can be left out from actual parameter lists if not followed by further actual parameters or shall be skipped by a dash in other cases (e.g. if there is a mandatory parameter following non-mandatory ones or an actual value to be passed to any of the non-mandatory parameters following the skipped one).
Details
Formal parameter lists

All parameterizable objects, i.e. types, templates, functions, altsteps and testcases may have non‑mandatory parameters along with mandatory parameters. Value and template kind formal parameters, port and timer parameters may be declared as non‑mandatory. To make a formal parameter non‑mandatory, a default value has to be assigned to it using the normal TTCN-3 assignment syntax. Its default value is assigned to a formal parameter each time the given parameterized object is called unless an actual value is passed to it in a given call.

Formal parameter default values shall be visible in the header of the parameterized object. I.e. they shall be literal values (incl. in-line templates), global definitions or component definitions when the header of the parameterized object has a runs on clause or expressions composed from them (i.e. a global constant may be used as a default value of an in parameter but a local constant shall not). This rule has the following consequences:
1) Only testcases, functions and altsteps with a runs on clause shall have non-mandatory inout and out value and template parameters. Moreover, default values of such parameters shall be component variables declared in the component type referred to by the runs on clause. These parameters have two operational modes:
  - if an actual parameter (variable) is passed in a call, they behave like mandatory parameters, i.e. the value of the parameter is returned to the variable on exit;
  - if no actual parameter is passed in a call, the initial value of an inout parameter will be the actual value of the related component variable, while out parameters are unbound when the object is called. Hence component variables used as the default values of inout parameters shall be initialized by the time the testcase, function or altstep is called or activated. The actual value of the parameter is returned to the component variable at exit.
2) Only testcases, functions and altsteps with a runs on clause shall have non-mandatory port and timer formal parameters. The default values shall only be ports and timers declared in the component type referred to by the runs on clause.

Functions called in expressions of default parameters shall have no side effect (the same restrictions apply as for functions called in the control part).

Examples:

const integer cg_int := 15;
type record MyRecord (in integer lower, in integer upper := 255) {
  integer field1 (lower .. upper) optional,
  octetstring field2 optional
}
type component MyComponentType {
  var integer v_compInt := 0;
  timer MyTimer := 10.0;
  timer My2ndTimer := 15.0
  port MyPortType P
  port MyPortType P1
}

template MyRecord (0) t_MyTemplate
  (in template integer par1 := 0, in template octetstring par2 := '00'O)
  := ( {?,-},{-,?},{ par1, par2 } );
function f_MyFunction
 (in integer par1 := cg_int,
  in charstring par2,
  template in charstring par3 := pattern “a?c”,
  inout integer par4 := v_compInt // only component variable is allowed here!
 ) runs on MyComponentType
{...}
altstep MyAltstep (MyPortType p_port := P, timer p_timer := MyTimer)
  runs on MyComponentType
{...}

altstep MyAltstep2 (MyPortType p_port := P, timer p_timer := MyTimer)
{...}
// causes a compile-time error because of the missing runs on clause
// (P and MyTimer are not visible)
altstep MyAltstep2 (MyPortType p_port := P2, timer p_timer := MyTimer)
  runs on MyComponentType {...}
// causes a compile-time error because no port P2 is declared in
// MyComponentType


Actual parameters

For all mandatory formal parameters an actual parameter must be provided. For non‑mandatory formal parameters the actual value may be omitted (and in this case the default value is used for the given call). Non-mandatory actual parameters shall be omitted explicitly by dash when followed by further actual parameters (either mandatory parameter(s) or non-mandatory parameter(s) with actual value(s)) or may be omitted implicitly - by leaving them out from the actual parameter list - when not followed by further actual parameter.
All other rules of parameterization remain as defined in the standard. Particularly, the order of elements in an actual parameter list shall be the same as the order in which they appear in the corresponding formal parameter list and the type of each actual parameter shall be compatible with the type of each corresponding formal parameter. In the case of nested parameter lists all parameterized entities specified as an actual parameter shall have their own parameters resolved in the actual parameter list.
When using parameterized types and templates without actual parameters (i.e. all formal parameters are non-mandatory and all are using their default values), an empty pair of parantheses shall be present.
Note, that local timers may be passed as actual parameters to non-mandatory timer parameters of functions and altsteps.
Examples:

type record MyRecord2 (in integer lower2 := 0, in integer upper2 := 255) {
  integer field1 (lower .. upper) optional,
  octetstring field2 optional
}

template MyRecord2 () t_MyTemplate2
  (in template integer par1 := 0, in template octetstring par2 := '00'O)
  := ( {?,-},{-,?},{ par1, par2 } );

...
var integer vl_int := 10;
template var charstring vtl_str := pattern "a??";
var default vl_def;
timer MyLocalTimer := 20.0;

P.send (t_MyTemplate2(1,'01'O));
P.send (t_MyTemplate2(1));
P.send (t_MyTemplate2(-,'FF'O));
P.send (t_MyTemplate2());

f_MyFunction(-,"abc");
f_MyFunction(5,"abc", vtl_str);
f_MyFunction(-,"abc", -, vl_int);
f_MyFunction(5,"abc", vtl_str, vl_int);
vl_def := activate (MyAltstep(P1,My2ndTimer));
vl_def := activate (MyAltstep(P1));
vl_def := activate (MyAltstep(P1,-));
vl_def := activate (MyAltstep(-,My2ndTimer));
vl_def := activate (MyAltstep(-,-));
vl_def := activate (MyAltstep()); // this is equivalent with the previous 
MyAltstep(-,MyLocalTimer);


	Other comments
	


Detailed amendments to the text of the standard

To be completed on publishing TTCN-3 core v3.2.1!
Changes in the BNF:

504. FormalValuePar ::= [(InParKeyword | InOutParKeyword | OutParKeyword)] Type ValueParIdentifier
                         AssignmentChar Expression
506. FormalPortPar ::= [InOutParKeyword] PortTypeIdentifier PortParIdentifier
                         AssignmentChar PortParIdentifier
508. FormalTimerPar ::= [InOutParKeyword] TimerKeyword TimerParIdentifier
                         AssignmentChar TimerParIdentifier
510. FormalTemplatePar ::= [(InParKeyword | OutParKeyword | InOutParKeyword )]
                           TemplateKeyword Type TemplateParIdentifier
                           AssignmentChar TemplateBody
180. FunctionActualPar ::= TimerRef |
                           TemplateInstance |
                           Port |
                           ComponentRef |
                           Dash
203. TestcaseActualPar ::= TemplateInstance | Dash
454. TypeActualPar ::= ConstantExpression | Dash
Note: BNF extensions are shown in italics.
